

Risk Attitude and Exponential Utility

A certainty equivalent is a certain payoff value which is equivalent, for the decision maker, to a particular payoff distribution. If the decision maker can determine his or her certainty equivalent for the payoff distribution of each strategy in a decision problem, then the optimal strategy is the one with the highest certainty equivalent.

The certainty equivalent, i.e., the minimum selling price for a payoff distribution, depends on the decision maker's personal attitude toward risk. A decision maker may be risk preferring, risk neutral, or risk avoiding.

If the terminal values are not regarded as extreme relative to the decision maker's total assets, if the decision maker will encounter other decision problems with similar payoffs, and if the decision maker has the attitude that he or she will "win some and lose some," then the decision maker's attitude toward risk may be described as risk neutral.

If the decision maker is risk neutral, the certainty equivalent of a payoff distribution is equal to its expected value. The expected value of a payoff distribution is calculated by multiplying each terminal value by its probability and summing the products.

If the terminal values in a decision situation are extreme or if the situation is "one-of-a-kind" so that the outcome has major implications for the decision maker, an expected value analysis may not be appropriate. Such situations may require explicit consideration of risk.

Unfortunately, it can be difficult to determine one's certainty equivalent for a complex payoff distribution. We can aid the decision maker by first determining his or her certainty equivalent for a simple payoff distribution and then using that information to infer the certainty equivalent for more complex payoff distributions.

A utility function, $U(x)$, can be used to represent a decision maker's attitude toward risk. The values or certainty equivalents, x , are plotted on the horizontal axis; utilities or expected utilities, u or $U(x)$, are on the vertical axis. You can use the plot of the function by finding a value on the horizontal axis, scanning up to the plotted curve, and looking left to the vertical axis to determine the utility.

A typical risk utility function might have the general shape shown below if you draw a smooth curve approximately through the points.


Figure 1 Typical Risk Utility Function



Since more value generally means more utility, the utility function is monotonically non-decreasing, and its inverse is well-defined. On the plot of the utility function, you locate a utility on the vertical axis, scan right to the plotted curve, and look down to read the corresponding value.

The concept of a payoff distribution, risk profile, gamble, or lottery is important for discussing utility functions. A payoff distribution is a set of payoffs, e.g., x_1 , x_2 , and x_3 , with corresponding probabilities, $P(X=x_1)$, $P(X=x_2)$, and $P(X=x_3)$. For example, a payoff distribution may be represented in decision tree form as shown below.

Figure 2 Payoff Distribution Probability Tree



The fundamental property of a utility function is that the utility of the certainty equivalent CE of a payoff distribution is equal to the expected utility of the payoffs, i.e.,

$$U(CE) = P(X=x_1) \cdot U(x_1) + P(X=x_2) \cdot U(x_2) + P(X=x_3) \cdot U(x_3).$$

It follows that if you compute the expected utility (EU) of a lottery,

$$EU = P(X=x_1)*U(x_1) + P(X=x_2)*U(x_2) + P(X=x_3)*U(x_3),$$

the certainty equivalent of the payoff distribution can be determined using the inverse of the utility function. That is, you locate the expected utility on the vertical axis, scan right to the plotted curve, and look down to read the corresponding certainty equivalent.

If a utility function has been determined, you can use this fundamental property to determine the certainty equivalent of any payoff distribution. Calculations for the Magnetic strategy in the DriveTek problem are shown below. First, using a plot of the utility function, locate each payoff x on the horizontal axis and determine the corresponding utility $U(x)$ on the vertical axis. Second, compute the expected utility EU of the lottery by multiplying each utility by its probability and summing the products. Third, locate the expected utility on the vertical axis and determine the corresponding certainty equivalent CE on the horizontal axis.

Figure 3 Calculations Using Risk Utility Function

| $P(X=x)$ | x | $U(x)$ | $P(X=x)*U(x)$ | |
|----------|-----------|--------|---------------|----|
| 0.50 | -\$50,000 | 0.00 | 0.0000 | |
| 0.15 | \$0 | 0.45 | 0.0675 | |
| 0.35 | \$120,000 | 0.95 | 0.3325 | |
| | | | 0.4000 | EU |
| | | | | |
| | | | -\$8,000 | CE |

Exponential Risk Utility

Instead of using a plot of a utility function, an exponential function may be used to represent risk attitude. The general form of the exponential utility function is

$$U(x) = A - B*EXP(-x/RT).$$

The risk tolerance parameter RT determines the curvature of the utility function reflecting the decision maker's attitude toward risk. Subsequent sections cover three methods for determining RT .

EXP is Excel's standard exponential function, i.e., $EXP(z)$ represents the value e raised to the power of z , where e is the base of the natural logarithms.

The parameters A and B determine scaling. After RT is determined, if you want to plot a utility function so that $U(High) = 1.0$ and $U(Low) = 0.0$, you can use the following formulas to determine the scaling parameters A and B .

$$A = EXP(-Low/RT) / [EXP(-Low/RT) - EXP(-High/RT)]$$

$$B = 1 / [\text{EXP}(-\text{Low}/RT) - \text{EXP}(-\text{High}/RT)]$$

The inverse function for finding the certainty equivalent CE corresponding to an expected utility EU is

$$CE = -RT * \text{LN}[(A - EU)/B],$$

where $\text{LN}(y)$ represents the natural logarithm of y .

After the parameters A , B , and RT have been determined, the exponential utility function and its inverse can be used to determine the certainty equivalent for any lottery.

Calculations for the Magnetic strategy in the DriveTek problem are shown in Figure 4.

Figure 4 Exponential Risk Utility Results

| | A | B | C | D | E |
|----|-----------------------------------|-----------|--------|-------------|-----------|
| 1 | Exponential Utility Inputs | | | | |
| 2 | RT | \$100,000 | | | |
| 3 | Low | -\$50,000 | | | |
| 4 | High | \$150,000 | | | |
| 5 | Computed | | | | |
| 6 | A | 1.1565 | | | |
| 7 | B | 0.7015 | | | |
| 8 | Payoff Distribution | | | | |
| 9 | P(X=x) | x | U(x) | P(X=x)*U(x) | |
| 10 | 0.50 | -\$50,000 | 0.0000 | 0.0000 | |
| 11 | 0.15 | \$0 | 0.4551 | 0.0683 | |
| 12 | 0.35 | \$120,000 | 0.9452 | 0.3308 | |
| 13 | | | | 0.3991 | EU |
| 14 | | | | | |
| 15 | | | | -\$7,676 | CE |

Computed values are displayed with four decimal places, but Excel's 15-digit precision is used in all calculations. For a decision maker with a risk tolerance parameter of \$100,000, the payoff distribution for the Magnetic strategy has a certainty equivalent of -\$7,676. That is, if the decision maker is facing the payoff distribution shown in A9:B12 in Figure 4, he or she would be willing to pay \$7,676 to be relieved of the obligation.

Formulas are shown in Figure 5. To construct the worksheet, enter the text in column A and the monetary values in column B. To define names, select A2:B4, and choose Insert | Name | Create. Similarly, select A6:B7, and choose Insert | Name | Create. Then enter the formulas in B6:B7. Enter formulas in C10 and D10, and copy down. Finally, enter the EU formula in D13 and the CE formula in D15. The defined names are absolute references by default.

Figure 5 Exponential Risk Utility Formulas

| | A | B | C | D | E |
|----|-----------------------------------|---|-----------------------------------|------------------------------------|----|
| 1 | Exponential Utility Inputs | | | | |
| 2 | RT | \$100,000 | | | |
| 3 | Low | -\$50,000 | | | |
| 4 | High | \$150,000 | | | |
| 5 | Computed | | | | |
| 6 | A | $=\text{EXP}(-\text{Low}/\text{RT})/(\text{EXP}(-\text{Low}/\text{RT})-\text{EXP}(-\text{High}/\text{RT}))$ | | | |
| 7 | B | $=1/(\text{EXP}(-\text{Low}/\text{RT})-\text{EXP}(-\text{High}/\text{RT}))$ | | | |
| 8 | Payoff Distribution | | | | |
| 9 | P(X=x) | x | U(x) | P(X=x)*U(x) | |
| 10 | 0.50 | -\$50,000 | $=A*B*\text{EXP}(-B10/\text{RT})$ | $=A10*C10$ | |
| 11 | 0.15 | \$0 | $=A*B*\text{EXP}(-B11/\text{RT})$ | $=A11*C11$ | |
| 12 | 0.35 | \$120,000 | $=A*B*\text{EXP}(-B12/\text{RT})$ | $=A12*C12$ | |
| 13 | | | | $=\text{SUM}(D10:D12)$ | EU |
| 14 | | | | | |
| 15 | | | | $=-\text{RT}*\text{LN}((A-D13)/B)$ | CE |

Figure 6 shows results for the same payoff distribution using a simplified form of the exponential risk utility function with $A = 1$ and $B = 1$. This function could be represented as $U(x) = 1 - \text{EXP}(-x/\text{RT})$ with inverse $\text{CE} = -\text{RT} * \text{LN}(1 - \text{EU})$. The utility and expected utility calculations are different, but the certainty equivalent is the same.


Figure 6 Simplified Exponential Risk Utility Results

| | A | B | C | D | E |
|----|-----------------------------------|-----------|---------|-------------|-----------|
| 1 | Exponential Utility Inputs | | | | |
| 2 | RT | \$100,000 | | | |
| 3 | Low | -\$50,000 | | | |
| 4 | High | \$150,000 | | | |
| 5 | Computed | | | | |
| 6 | A | 1 | | | |
| 7 | B | 1 | | | |
| 8 | Payoff Distribution | | | | |
| 9 | P(X=x) | x | U(x) | P(X=x)*U(x) | |
| 10 | 0.50 | -\$50,000 | -0.6487 | -0.3244 | |
| 11 | 0.15 | \$0 | 0.0000 | 0.0000 | |
| 12 | 0.35 | \$120,000 | 0.6988 | 0.2446 | |
| 13 | | | | -0.0798 | EU |
| 14 | | | | | |
| 15 | | | | -\$7,676 | CE |

Approximate Risk Tolerance

The value of the risk tolerance parameter RT is approximately equal to the maximum value of Y for which the decision maker is willing to accept a payoff distribution with equally-likely payoffs of \$Y and -\$Y/2 instead of accepting \$0 for certain.

Figure 7 Approximate Risk Tolerance




For example, in a personal decision, you may be willing to play the game shown in Figure 7 with equally-likely payoffs of \$100 and –\$50, but you might not play with payoffs of \$100,000 and –\$50,000. As the better payoff increases from \$100 to \$100,000 (and the corresponding worse payoff increases from –\$50 to –\$50,000), you reach a value where you are indifferent between playing the game and receiving \$0 for certain. At that point, the value of the better payoff is an approximation of RT for an exponential risk utility function describing your risk attitude.

In a business decision for a small company, the company may be willing to play the game with payoffs of \$200,000 and –\$100,000 but not with payoffs of \$20,000,000 and –\$10,000,000. Somewhere between a better payoff of \$200,000 and \$20,000,000, the company would be indifferent between playing the game and not playing, thereby determining the approximate RT for their business decision.

Exact Risk Tolerance Using Goal Seek and Solver

A simple payoff distribution, called a risk attitude assessment lottery, may be used to determine the decision maker's attitude toward risk. This lottery has equal probability of obtaining each of the two payoffs. It is good practice to use a better payoff at least as large as the highest payoff in the decision problem and a worse payoff as small as or smaller than the lowest payoff. In any case, the payoffs should be far enough apart that the decision maker perceives a definite difference in the two outcomes. Three values must be specified for the fifty-fifty lottery: the Better payoff, the Worse payoff, and the Certainty Equivalent, as shown in Figure 8.

Figure 8 Risk Attitude Assessment Lottery



According to the fundamental property of a risk utility function, the utility of the certainty equivalent equals the expected utility of the lottery, so the three values are related as follows.

$$U(\text{CertEquiv}) = 0.5 * U(\text{BetterPayoff}) + 0.5 * U(\text{WorsePayoff})$$

If you use the general form for an exponential utility function with parameters A, B, and RT, and if you simplify terms, it follows that RT must satisfy the following equation.

$$\text{Exp}(-\text{CertEquiv}/\text{RT}) = 0.5 * \text{Exp}(-\text{BetterPayoff}/\text{RT}) + 0.5 * \text{Exp}(-\text{WorsePayoff}/\text{RT})$$

Given the values for CE, Better, and Worse, you could use trial-and-error to find the value of RT that exactly satisfies the equation. In Excel you can use Goal Seek or Solver by creating a worksheet like Figure 9.

Enter the text in column A. Enter the assessment lottery values in B2:B4. Enter a tentative RT value in B6. Select A2:B4, and use Insert | Name | Create; repeat for A6:B6 and A8:B9. Note that the parentheses symbol is not allowed in a defined name, so Excel changes U(CE) to U_CE and EU(Lottery) to EU_Lottery.

Figure 9 Formulas for Risk Tolerance Search

| | A | B | C | D | E | F | G |
|----|--------------------------|---|---|---|---|---|---|
| 1 | Assessment Inputs | | | | | | |
| 2 | WorsePayoff | -\$50,000 | | | | | |
| 3 | CertEquiv | \$30,000 | | | | | |
| 4 | BetterPayoff | \$150,000 | | | | | |
| 5 | Changing Cell | | | | | | |
| 6 | RT | \$200,000 | | | | | |
| 7 | Computed Values | | | | | | |
| 8 | U(CE) | =EXP(-CertEquiv/RT) | | | | | |
| 9 | EU(Lottery) | =0.5*EXP(-BetterPayoff/RT)+0.5*EXP(-WorsePayoff/RT) | | | | | |
| 10 | Target Cell | | | | | | |
| 11 | Difference | =U_CE-EU_Lottery | | | | | |
| 12 | | | | | | | |

Figure 10 Tentative Values for Risk Tolerance Search

| | A | B |
|----|--------------------------|-----------|
| 1 | Assessment Inputs | |
| 2 | WorsePayoff | -\$50,000 |
| 3 | CertEquiv | \$30,000 |
| 4 | BetterPayoff | \$150,000 |
| 5 | Changing Cell | |
| 6 | RT | \$200,000 |
| 7 | Computed Values | |
| 8 | U(CE) | 0.860708 |
| 9 | EU(Lottery) | 0.878196 |
| 10 | Target Cell | |
| 11 | Difference | -0.01749 |

Figure 10 shows tentative values for the search. From the Tools menu, choose Goal Seek. In the Goal Seek dialog box, enter B11, 0, and B6. If you point to cells, the reference appears in the edit box as an absolute reference, as shown in Figure 11. Click OK.

Figure 11 Goal Seek Dialog Box

The screenshot shows the 'Goal Seek' dialog box with the following fields:

- Set cell:** \$B\$11
- To value:** 0
- By changing cell:** \$B\$6

Buttons: OK, Cancel

The Goal Seek Status dialog box shows that a solution has been found. Click OK. The worksheet appears as shown in Figure 12.

Figure 12 Results of Goal Seek Search

| | A | B |
|----|--------------------------|-----------|
| 1 | Assessment Inputs | |
| 2 | WorsePayoff | -\$50,000 |
| 3 | CertEquiv | \$30,000 |
| 4 | BetterPayoff | \$150,000 |
| 5 | Changing Cell | |
| 6 | RT | \$242,357 |
| 7 | Computed Values | |
| 8 | U(CE) | 0.88357 |
| 9 | EU(Lottery) | 0.883828 |
| 10 | Target Cell | |
| 11 | Difference | -0.00026 |

The difference between U(CE) and EU(Lottery) is not exactly zero. If you start at \$250,000, the Goal Seek converges to a difference of $-6.2\text{E}-05$ or 0.000062, which is closer to zero, resulting in a RT of \$243,041.

If extra precision is needed, use Solver. With Solver's default settings, the difference is $2.39\text{E}-08$ with RT equal to \$243,261. If you change the precision from 0.000001 to 0.00000001 or an even smaller value in Solver's Options, the difference will be even closer to zero.

Exact Risk Tolerance Using RiskTol.xla

The Goal Seek and Solver methods for determining the risk tolerance parameter RT yield static results. For a dynamic result, use the risktol.xla add-in function. A major advantage of risktol.xla is that it facilitates sensitivity analysis. Whenever an input to the function changes, the result is recalculated. The function syntax is

RISKTOL(*WorsePayoff*,*CertEquiv*,*BetterPayoff*,*BetterProb*).

When you open the risktol.xla file, the function is added to the Math & Trig function category list.

The function returns a very precise value of the risk tolerance parameter for an exponential utility function. The result is consistent with CertEquiv as the decision maker's certainty equivalent for a two-payoff assessment lottery with payoffs WorsePayoff and BetterPayoff, with probability BetterProb of obtaining BetterPayoff and probability $1 - \text{BetterProb}$ of obtaining WorsePayoff.

In case of an error, the RISKTOL function returns:

#N/A if there are too few or too many arguments. The first three arguments (WorsePayoff, CertEquiv, and BetterPayoff) are required; the fourth argument (BetterProb) is optional, with default value 0.5.

#VALUE! if WorsePayoff \geq CertEquiv, or CertEquiv \geq Better Payoff, or BetterProb (if specified) ≤ 0 or ≥ 1 .

#NUM! if the search procedure fails to converge.

In Figure 13, the text in cells A2:A4 has been used as defined names for cells B2:B4, and the text in cell A6 is the defined name for cell B6, as shown in the name box. After opening the risktol.xla file, enter the function name and arguments, as shown in the formula bar. If one of the three inputs change, the result in cell B6 is recalculated.

Figure 13 Exact Risk Tolerance Using RiskTol.xla

| | RT | | =RISKTOL(BetterPayoff,CertEquiv,WorsePayoff) | | | |
|---|--------------------------|-----------|--|---|---|---|
| | A | B | C | D | E | F |
| 1 | Assessment Inputs | | | | | |
| 2 | WorsePayoff | \$150,000 | | | | |
| 3 | CertEquiv | \$30,000 | | | | |
| 4 | BetterPayoff | -\$50,000 | | | | |
| 5 | Risk Tolerance | | | | | |
| 6 | RT | \$243,261 | | | | |
| 7 | | | | | | |

Exponential Utility and TreePlan

TreePlan's default is to rollback the tree using expected values. If you choose to use exponential utilities in TreePlan's Options dialog box, TreePlan will redraw the decision tree diagram with formulas for computing the utility and certainty equivalent at each node. For the Maximize option, the rollback formulas are $U = A - B * \exp(-X/RT)$ and $CE = -\ln((A - EU)/B) * RT$, where X and EU are cell references. For the Minimize option, the formulas are $U = A - B * \exp(X/RT)$ and $CE = \ln((A - EU)/B) * RT$.

TreePlan uses the name RT to represent the risk tolerance parameter of the exponential utility function. The names A and B determine scaling. If the names A, B, and RT don't exist on the worksheet when you choose to use exponential utility, they are initially defined as A=1, B=1, and RT=999999999999. You can redefine the names using the Insert | Name | Define or Insert | Name | Create commands.

To plot the utility curve, enter a list of X values in a column on the left, and enter the formula $=A - B * \exp(-X/RT)$ in a column on the right, where X is a reference to the corresponding cell on the left. Select the values in both columns, and use the ChartWizard to develop an XY (Scatter) chart.

If RT is specified using approximate risk tolerance values, you can perform sensitivity analysis by (1) using the defined name RT for a cell, (2) constructing a data table with a list of possible RT values and an appropriate output formula (usually a choice indicator at a decision node or a certainty equivalent), and (3) specifying the RT cell as the input cell in the Data Table dialog box.

Exponential Utility and RiskSim

After using RiskSim to obtain model output results, select the column containing the Sorted Data, copy to the clipboard, select a new sheet, and paste. Alternatively, you can use the unsorted values, and you can also do the following calculations on the original sheet containing the model results. This example uses only ten iterations; 500 or 1,000 iterations are more appropriate.

Use one of the methods described previously to specify values of RT, A, and B. Since the model output values shown in Figures 14 and 15 range from approximately \$14,000 to \$176,000, the utility function is defined for a range from worse payoff \$0 to better payoff \$200,000. RT was determined using risktol.xls with a risk-seeking certainty equivalent of \$110,000.

To obtain the utility of each model output value in cells A2:A11, select cell B2, and enter the formula $=A-B*EXP(-A2/RT)$. Select cell B2, click the fill handle in the lower right corner of the cell and drag down to cell B11. Enter the formulas in cells A13:C13 and the labels in row 14.

Figure 14 Risk Utility Formulas for RiskSim

| | A | B | C |
|----|--------------------|---------------------|---------------------|
| 1 | Sorted Data | Utility | |
| 2 | 14229.56 | $=A-B*EXP(-A2/RT)$ | |
| 3 | 32091.92 | $=A-B*EXP(-A3/RT)$ | |
| 4 | 51091.48 | $=A-B*EXP(-A4/RT)$ | |
| 5 | 66383.79 | $=A-B*EXP(-A5/RT)$ | |
| 6 | 69433.32 | $=A-B*EXP(-A6/RT)$ | |
| 7 | 87322.23 | $=A-B*EXP(-A7/RT)$ | |
| 8 | 95920.93 | $=A-B*EXP(-A8/RT)$ | |
| 9 | 135730.71 | $=A-B*EXP(-A9/RT)$ | |
| 10 | 154089.36 | $=A-B*EXP(-A10/RT)$ | |
| 11 | 175708.87 | $=A-B*EXP(-A11/RT)$ | |
| 12 | | | |
| 13 | $=AVERAGE(A2:A11)$ | $=AVERAGE(B2:B11)$ | $=LN((A-B13)/B)*RT$ |
| 14 | Exp. Value | Exp.Util. | CE |

Figure 15 Risk Utility Results for RiskSim

| | A | B | C |
|----|-------------|-----------|-----------|
| 1 | Sorted Data | Utility | |
| 2 | \$ 14,230 | 0.05862 | |
| 3 | \$ 32,092 | 0.13462 | |
| 4 | \$ 51,091 | 0.21851 | |
| 5 | \$ 66,384 | 0.28841 | |
| 6 | \$ 69,433 | 0.30260 | |
| 7 | \$ 87,322 | 0.38767 | |
| 8 | \$ 95,921 | 0.42966 | |
| 9 | \$ 135,731 | 0.63382 | |
| 10 | \$ 154,089 | 0.73363 | |
| 11 | \$ 175,709 | 0.85600 | |
| 12 | | | |
| 13 | \$ 88,200 | 0.40435 | \$ 90,757 |
| 14 | Exp. Value | Exp.Util. | CE |

Michael R. Middleton
School of Business and Management
University of San Francisco
2130 Fulton Street
San Francisco, CA 94117
middleton@usfca.edu

October 18, 2001